Free Lie Algebras Routines

Lazy Evaluation Version

Pensieve header: A free-Lie calculator, lazy evaluation version.

Global Definitions

```
$LieSeriesShowDegree = 3; $LieSeriesCompareDegree = 3;
```

NonCommutativeMultiply

```
Unprotect[NonCommutativeMultiply];
x_ ** 0 = 0; 0 ** y_ = 0;
(c_ * x_AW) ** y_ := Expand[c (x ** y)];
x_ ** (c_ * y_AW) := Expand[c (x ** y)];
x_Plus ** y_ := (# ** y) & /@ x;
x_ ** y_Plus := (x ** #) & /@ y;
AW[w1_String] ** AW[w2_String] := AW[w1 <> w2];
```

Words and Lyndon Words

A Lyndon word is a word lexicographically smaller than all of its proper right factors; see http://katlas.math.toronto.edu/drorbn/AcademicPensieve/Projects/FreeLie/index.html

```
LyndonQ[AW[w_String]] := And @@ (
    OrderedQ[{w, #}] & /@ Table[StringDrop[w, i], {i, 1, StringLength[w] - 1}]
   );
AllWords[0, _List] = {AW[""]};
AllWords[n_ /; n > 0, ab_List] := AllWords[n, ab] = AW /@ Flatten[Outer[
      StringJoin[#1, #2] &,
      First /@ AllWords[n - 1, ab],
      ab
     ]];
AllLyndonWords[n_Integer, ab_List] := LW @@@ Select[AllWords[n, ab], LyndonQ];
AllLyndonWords[{n_}, ab_List] := Join @@ Table[AllLyndonWords[k, ab], {k, n}];
LyndonFactorization[LW[w_String] /; StringLength[w] == 1] := LW[w];
LyndonFactorization[LW[w_String] /; StringLength[w] > 1] := Module[
   {rf},
   rf = First[Sort[Table[StringDrop[w, i], {i, 1, StringLength[w] - 1}]]];
   LW /@ {StringDrop[w, -StringLength[rf]], rf}
  ];
LW[s_Symbol] := LW[ToString[s]];
LW[LW[w_]] := LW[w];
LW /: LW[x_] ≤ LW[y_] := OrderedQ[{x, y}];
LW /: x_LW ≥ y_LW := y ≤ x;
LW /: x_LW > y_LW := ! (x ≤ y);
LW /: x_LW < y_LW := ! (y ≤ x);
Format[LW[w_], StandardForm] := Defer[⟨w⟩];
⟨w__⟩ := LW[w];
LW[is__Integer] := LW[
   StringJoin @@ (StringTake["1234567890abcdefghijklmnopqrstuvwxyz", {#}] & /@ {is})];
Deg[LW[x_]] := StringLength[x];
```

```
{LyndonQ[AW@"abba"], LyndonQ[AW@"ababb"]}
```

```
{False, True}
```

```
{AllWords[3, {"1", "2"}], AllLyndonWords[{3}, {"1", "2"}]}
```

```
{{AW[111], AW[112], AW[121], AW[122], AW[211], AW[212], AW[221], AW[222]},
 {⟨1⟩, ⟨2⟩, ⟨12⟩, ⟨112⟩, ⟨122⟩}}
```

```
Table[Length[AllLyndonWords[k, {"1", "2"}]], {k, 10}]
```

```
{2, 1, 2, 3, 6, 9, 18, 30, 56, 99}
```

```
Table[Length[AllLyndonWords[k, {"1", "2", "3"}]], {k, 10}]
```

```
{3, 3, 8, 18, 48, 116, 312, 810, 2184, 5880}
```

```
TreeForm[LW["12122"] //. w_LW :> LyndonFactorization[w] /. LW[w_] :> w]
```



The Bracket for Lie Elements

```
b[0, _] = 0; b[_, 0] = 0;
b[c_ * (x_AW | x_LW), y_] := Expand[c b[x, y]];
b[x_, c_ * (y_AW | y_LW)] := Expand[c b[x, y]];
b[x_Plus, y_] := b[#, y] & /@ x;
b[x_, y_Plus] := b[x, #] & /@ y;
b[w_AW, z_AW] := w ** z - z ** w;
b[w_LW, z_LW] := LWBracket[w, z];
ad[x_][y_] := b[x, y];
```

```
LWBracket[w_LW, z_LW] := Which[
   w === z, 0,
   z < w, Expand[-b[z, w]],
   Deg[w] == 1, LW[First[w] <> First[z]],
   True, Module[{x, y},
    {x, y} = LyndonFactorization[w];
    If[y ≥ z,
     LW[First[w] <> First[z]],
     LWBracket[w, z] = b[x, LWBracket[y, z]] + b[LWBracket[x, z], y]
    ]
   ]
  ];
```

```
b[LW["112"], LW["122"]]
```

```
⟨112122⟩ + ⟨112212⟩
```

```
Outer[b, AllLyndonWords[{3}, {"1", "2"}],
  AllLyndonWords[{3}, {"1", "2"}]] // MatrixForm
```

0	⟨12⟩	⟨112⟩	⟨1112⟩	⟨1122⟩
-⟨12⟩	0	-⟨122⟩	-⟨1122⟩	-⟨1222⟩
-⟨112⟩	⟨122⟩	0	-⟨11212⟩	⟨12122⟩
-⟨1112⟩	⟨1122⟩	⟨11212⟩	0	⟨112122⟩ + ⟨112212⟩
-⟨1122⟩	⟨1222⟩	-⟨12122⟩	-⟨112122⟩ - ⟨112212⟩	0

```
Union[Flatten[Outer[(b[#1, #2] + b[#2, #1]) &,
   AllLyndonWords[{6}, {"1", "2"}], AllLyndonWords[{6}, {"1", "2"}]
  ]]]
```

```
{0}
```

```
Outer[(b[#1, b[#2, #3]] + b[#2, b[#3, #1]] + b[#3, b[#1, #2]]) &,
   AllLyndonWords[{5}, {"1", "2"}],
   AllLyndonWords[{5}, {"1", "2"}], AllLyndonWords[{5}, {"1", "2"}]
  ] // Flatten //
 Union
```

```
{0}
```

LieSeries

```
LieSeries[ser_Symbol][e___] := ser[e];
Format[LieSeries[s_Symbol], StandardForm] :=
  LS @@ Table[s[d], {d, $LieSeriesShowDegree}];
MakeLieSeries[s_LieSeries] := s;
MakeLieSeries[expr_] :=
  MakeLieSeries[expr] = MakeLieSeries[Unique[MakeLieSeries], expr];
MakeLieSeries[ser_Symbol, expr_] := (
   ser[] = Hold[MakeLieSeries[ser, expr]];
   ser[d_Integer] := ser[d] = expr /. w_LW /; Deg[w] ≠ d → 0;
   LieSeries[ser]
  );
s1_LieSeries ≡ s2_LieSeries :=
  And @@ ((s1[#] == s2[#]) & /@ Range[$LieSeriesCompareDegree]);
```

```
Print /@ {ts1 = ⟨"1122"⟩ // MakeLieSeries, ts1[], ts1 /@ Range[6]};
```

```
LS[0, 0, 0]
```

```
Hold[MakeLieSeries[MakeLieSeries$554, ⟨1122⟩]]
```

```
{0, 0, 0, ⟨1122⟩, 0, 0}
```

```
b[s1_LieSeries, s2_LieSeries] := b[s1, s2] = Module[{ser},
    ser = Unique[b];
    ser[] = Hold[b[s1, s2]];
    ser[d_Integer] := ser[d] = Sum[
       b[s1[k], s2[d - k]],
       {k, 1, d - 1}
      ];
    LieSeries[ser]
   ];
b[s_LieSeries, y_] := b[s, MakeLieSeries[y]];
b[x_, s_LieSeries] := b[MakeLieSeries[x], s];
```

```
{ts2 = ⟨"122"⟩ + ⟨"11122"⟩ // MakeLieSeries, ts3 = b[ts1, ts2], ts3[], ts3 /@ Range[10]}
```

```
{LieSeries[MakeLieSeries$545], LieSeries[b$547],
 Hold[b[LieSeries[MakeLieSeries$543], LieSeries[MakeLieSeries$545]]],
 {0, 0, 0, 0, 0, 0, ⟨1122122⟩, 0, -⟨111221122⟩, 0}}
```

```
LieSeries /: EulerE[s_LieSeries] := Module[{ser},
   ser = Unique[EulerE];
   ser[] = Hold[EulerE[s]];
   ser[d_Integer] := ser[d] = Expand[d * s[d]];
   LieSeries[ser]
  ];
```

```
{ts4 = EulerE[ts3], ts4[], ts4 /@ Range[10]}
```

```
{LieSeries[EulerE$554], Hold[EulerE[LieSeries[b$547]]],
 {0, 0, 0, 0, 0, 0, 7 〈1122122〉, 0, -9 〈111221122〉, 0}}
```

```
adPower[0, x_LieSeries][ψ_LieSeries] := adPower[0, x][ψ] = Module[{ser},
    ser = Unique[adPower];
    ser[] = Hold[adPower[0, x][ψ]];
    ser[d_Integer] := ser[d] = ψ[d];
    LieSeries[ser]
   ];
adPower[n_Integer, x_LieSeries][ψ_LieSeries] := adPower[n, x][ψ] = Module[{ser},
    ser = Unique[adPower];
    ser[] = Hold[adPower[n, x][ψ]];
    ser[d_Integer] := ser[d] =  b[x, adPower[n - 1, x][ψ]][d];
    LieSeries[ser]
   ];
adSeries[f_, x_LieSeries][ψ_LieSeries] := adSeries[f, x][ψ] = Module[{ser},
    ser = Unique[adSeries];
    ser[] = Hold[adSeries[f, x][ψ]];
    ser[d_Integer] := ser[d] = Module[{c},
       Expand[Sum[
         c = SeriesCoefficient[f, {ad, 0, k}];
         If[c == 0, 0, c * adPower[k, x][ψ][d]],
         {k, 0, d - 1}
        ]]
      ];
    LieSeries[ser]
   ];
adSeries[f_, x_][ψ_] := adSeries[f, MakeLieSeries[x]][MakeLieSeries[ψ]];
Ad[x_] := adSeries[E^(-ad), x];
```

```
{xs = MakeLieSeries[LW["x"]], ys = MakeLieSeries[LW["y"]],
 ts5 = adPower[0, xs][ys], ts5[], ts5 /@ Range[5]}
```

```
{LieSeries[MakeLieSeries$98], LieSeries[MakeLieSeries$99], LieSeries[adPower$101],
 Hold[adPower[0, LieSeries[MakeLieSeries$98]][LieSeries[MakeLieSeries$99]]],
 {〈y〉, 0, 0, 0, 0}}
```

```
adPower[3, xs][ys] /@ Range[5]
```

```
{0, 0, 0, 〈xxxy〉, 0}
```

```
{adSeries[E^(-ad), xs][ys] /@ Range[5], adSeries[E^(-ad), ys][xs] /@ Range[5]}
```

$$\left\{\left\{\langle y\rangle, -\langle xy\rangle, \frac{\langle xxy\rangle}{2}, -\frac{\langle xxxy\rangle}{6}, \frac{\langle xxxxy\rangle}{24}\right\}, \left\{\langle x\rangle, \langle xy\rangle, \frac{\langle xyy\rangle}{2}, \frac{\langle xyyy\rangle}{6}, \frac{\langle xyyyy\rangle}{24}\right\}\right\}$$

```
Ad[xs][ys][5]
```

```
⟨xxxxy⟩
───────
  24
```

```
Ad[xs][ys][]
```

```
Hold[adSeries[ⅇ^-ad, LieSeries[MakeLieSeries$98]][LieSeries[MakeLieSeries$99]]]
```

LieMorphism

```
LieMorphism[mor_][es___] := mor[es];
LieMorphism[rules_List] :=
  LieMorphism[rules] = LieMorphism[Unique[LieMorphism], rules];
LieMorphism[mor_Symbol, rules_List] := (
   mor[] = Hold[LieMorphism[mor, rules]];
   (mor[w_LW] /; Deg[w] == 1) := (mor[w] = MakeLieSeries[w /. rules]);
   mor[w_LW] := (mor[w] = b @@ (mor /@ LyndonFactorization[w]));
   mor[s_LieSeries] := mor[s] = Module[{ser},
      ser = Unique[LieMorphismOnLieSeries];
      ser[] = Hold[mor[s]];
      ser[d_] := ser[d] = Sum[
         mor[s[k]][d],
         {k, 1, d}
        ];
      LieSeries[ser]
     ];
   mor[expr_][d_] := Expand[expr /. w_LW :> mor[w][d]];
   LieMorphism[mor]
  );
```

```
{lm1 = LieMorphism[{LW["x"] → Ad[LW["y"]][LW["x"]]}], lm1[],
 lm1[LW["y"]], lm1[LW["x"]], lm1[LW["x"]][4], lm1[⟨"xxy"⟩], lm1[⟨"xxy"⟩][8]
}
```

```
{LieMorphism[LieMorphism$103],
 Hold[LieMorphism[LieMorphism$103, {⟨x⟩ → LieSeries[adSeries$102]}]],
                                    ⟨xyyy⟩
 ⟨y⟩, LieSeries[adSeries$102],  ──────,
                                      6
                      ⟨xxyyyyyy⟩     ⟨xyxyyyyy⟩     ⟨xyyxyyyy⟩
 LieSeries[b$132],  ────────── + ────────── + ──────────}
                         120            30             24
```

StableApply

```
StableApply[mor_LieMorphism, s_LieSeries] := StableApply[mor, s] = Module[{ser},
    ser = Unique[StableApply];
    ser[] = Hold[StableApply[mor, s]];
    ser[d_] := ser[d] = Nest[mor, s, d][d];
    (* ser[d_] := ser[d] = Module[{mm},
        mm=FixedPoint[mor, s, SameTest → (#1[d]==#2[d]&)];
        mm[d]
       ]; *)
    LieSeries[ser]
   ];
```

BCH

```
BCHBase = Module[{bch},
   bch = Unique["BCHBase"];
   bch[] = Hold[BCHBase];
   bch[1] = ⟨"x"⟩ + ⟨"y"⟩;
   bch[d_Integer] := bch[d] = Expand[Plus[
        adSeries[E^(-ad), MakeLieSeries[⟨"y"⟩]][MakeLieSeries[⟨"x"⟩]][d],
        -adSeries[(1 - E^(-ad)) / ad - 1, LieSeries[bch]][EulerE[LieSeries[bch]]][d]
      ] / d];
   LieSeries[bch]
  ];
BCH[x_, y_] := LieMorphism[{LW["x"] → x, LW["y"] → y}][BCHBase];
```

```
{BCHBase, BCHBase[], BCHBase[8]}
```

$$\Big\{\text{BCHBase3}\Big[\langle\text{x}\rangle+\langle\text{y}\rangle,\ \frac{\langle\text{xy}\rangle}{2},\ \frac{\langle\text{xxy}\rangle}{12}+\frac{\langle\text{xyy}\rangle}{12}\Big],\ \text{Hold}[\text{BCHBase}],\ \frac{\langle\text{xxxxxxyy}\rangle}{60\,480}-\frac{\langle\text{xxxxxyxy}\rangle}{15\,120}-\frac{\langle\text{xxxxxyyy}\rangle}{10\,080}+\frac{\langle\text{xxxxyxxy}\rangle}{20\,160}-\frac{\langle\text{xxxxyxyy}\rangle}{20\,160}+\frac{\langle\text{xxxxyyxy}\rangle}{2520}+\frac{23\,\langle\text{xxxxyyyy}\rangle}{120\,960}+\frac{\langle\text{xxxyxxyy}\rangle}{4032}-\frac{\langle\text{xxxyxyxy}\rangle}{10\,080}+\frac{13\,\langle\text{xxxyxyyy}\rangle}{30\,240}+\frac{\langle\text{xxxyyxyy}\rangle}{20\,160}-\frac{\langle\text{xxxyyyxy}\rangle}{3024}-\frac{\langle\text{xxxyyyyy}\rangle}{10\,080}+\frac{\langle\text{xxyxyxyy}\rangle}{2520}-\frac{\langle\text{xxyxyyyy}\rangle}{4032}-\frac{\langle\text{xxyyxyyy}\rangle}{10\,080}+\frac{\langle\text{xxyyyyyy}\rangle}{60\,480}\Big\}$$

$$\Big\{\text{LieSeries}[\text{BCHBase3}],\ \text{Hold}[\text{BCHBase}],\ \frac{\langle\text{xxxxxxyy}\rangle}{60\,480}-\frac{\langle\text{xxxxxyxy}\rangle}{15\,120}-\frac{\langle\text{xxxxxyyy}\rangle}{10\,080}+\frac{\langle\text{xxxxyxxy}\rangle}{20\,160}-\frac{\langle\text{xxxxyxyy}\rangle}{20\,160}+\frac{\langle\text{xxxxyyxy}\rangle}{2520}+\frac{23\,\langle\text{xxxxyyyy}\rangle}{120\,960}+\frac{\langle\text{xxxyxxyy}\rangle}{4032}-\frac{\langle\text{xxxyxyxy}\rangle}{10\,080}+\frac{13\,\langle\text{xxxyxyyy}\rangle}{30\,240}+\frac{\langle\text{xxxyyxyy}\rangle}{20\,160}-\frac{\langle\text{xxxyyyxy}\rangle}{3024}-\frac{\langle\text{xxxyyyyy}\rangle}{10\,080}+\frac{\langle\text{xxyxyxyy}\rangle}{2520}-\frac{\langle\text{xxyxyyyy}\rangle}{4032}-\frac{\langle\text{xxyyxyyy}\rangle}{10\,080}+\frac{\langle\text{xxyyyyyy}\rangle}{60\,480}\Big\}$$

```
{BCH[LW["y"], LW["z"]], BCH[LW["y"], LW["z"]][6]}
```

$$\Big\{\text{LieSeries}[\text{LieMorphismOnLieSeries\$101}],\ -\frac{\langle\text{yyyyzz}\rangle}{1440}+\frac{\langle\text{yyyzyz}\rangle}{720}+\frac{\langle\text{yyyzzz}\rangle}{360}+\frac{\langle\text{yyzyzz}\rangle}{240}-\frac{\langle\text{yyzzzz}\rangle}{1440}\Big\}$$

```
{
  t1 = BCH[LW["x"], BCH[LW["y"], LW["z"]]],
  t2 = BCH[BCH[LW["x"], LW["y"]], LW["z"]],
  t1 == t2,
  Table[t1[d] == t2[d], {d, 10 }]
 } // Timing
```

$$\Big\{7.987,\ \Big\{\mathrm{LS}\Big[\langle x\rangle+\langle y\rangle+\langle z\rangle,\ \frac{\langle xy\rangle}{2}+\frac{\langle xz\rangle}{2}+\frac{\langle yz\rangle}{2},\ \frac{\langle xxy\rangle}{12}+\frac{\langle xxz\rangle}{12}+\frac{\langle xyy\rangle}{12}+\frac{\langle xyz\rangle}{3}+\frac{\langle xzy\rangle}{6}+\frac{\langle xzz\rangle}{12}+\frac{\langle yyz\rangle}{12}+\frac{\langle yzz\rangle}{12}\Big],\ \mathrm{LS}\Big[\langle x\rangle+\langle y\rangle+\langle z\rangle,\ \frac{\langle xy\rangle}{2}+\frac{\langle xz\rangle}{2}+\frac{\langle yz\rangle}{2},\ \frac{\langle xxy\rangle}{12}+\frac{\langle xxz\rangle}{12}+\frac{\langle xyy\rangle}{12}+\frac{\langle xyz\rangle}{3}+\frac{\langle xzy\rangle}{6}+\frac{\langle xzz\rangle}{12}+\frac{\langle yyz\rangle}{12}+\frac{\langle yzz\rangle}{12}\Big],$$

$$\mathrm{LS}\Big[\langle x\rangle+\langle y\rangle+\langle z\rangle,\ \frac{\langle xy\rangle}{2}+\frac{\langle xz\rangle}{2}+\frac{\langle yz\rangle}{2},\ \frac{\langle xxy\rangle}{12}+\frac{\langle xxz\rangle}{12}+\frac{\langle xyy\rangle}{12}+\frac{\langle xyz\rangle}{3}+\frac{\langle xzy\rangle}{6}+\frac{\langle xzz\rangle}{12}+\frac{\langle yyz\rangle}{12}+\frac{\langle yzz\rangle}{12}\Big] == \mathrm{LS}\Big[\langle x\rangle+\langle y\rangle+\langle z\rangle,\ \frac{\langle xy\rangle}{2}+\frac{\langle xz\rangle}{2}+\frac{\langle yz\rangle}{2},\ \frac{\langle xxy\rangle}{12}+\frac{\langle xxz\rangle}{12}+\frac{\langle xyy\rangle}{12}+\frac{\langle xyz\rangle}{3}+\frac{\langle xzy\rangle}{6}+\frac{\langle xzz\rangle}{12}+\frac{\langle yyz\rangle}{12}+\frac{\langle yzz\rangle}{12}\Big],$$

$$\{\mathrm{True},\mathrm{True},\mathrm{True},\mathrm{True},\mathrm{True},\mathrm{True},\mathrm{True},\mathrm{True},\mathrm{True},\mathrm{True}\}\Big\}\Big\}$$

```
ι[w_LW] /; Deg[w] == 1 := AW @@ w;
ι[w_LW] := ι[w] = b @@ (ι /@ LyndonFactorization[w]);
ι[s_LieSeries] := Prepend[ι /@ (ASeries @@ s), 0];
ι[expr_] := Expand[expr /. w_LW ⧴ ι[w]];
```

```
t1 = ι[BCH[3]]
```

$$\mathrm{ASeries}\Big[0,\ \mathrm{AW}[x]+\mathrm{AW}[y],\ \frac{\mathrm{AW}[xy]}{2}-\frac{\mathrm{AW}[yx]}{2},\ \frac{\mathrm{AW}[xxy]}{12}-\frac{\mathrm{AW}[xyx]}{6}+\frac{\mathrm{AW}[xyy]}{12}+\frac{\mathrm{AW}[yxx]}{12}-\frac{\mathrm{AW}[yxy]}{6}+\frac{\mathrm{AW}[yyx]}{12}\Big]$$

```
ASeries /: Expand[s_ASeries] := Expand /@ s;
ASeries /: Plus[ss__ASeries] := Module[
   {l = Min[Length /@ {ss}]},
   ASeries @@ Total[Take[List @@ #, l] & /@ {ss}]
  ];
ASeries /: c_ * s_ASeries := Expand[c * #] & /@ s;
s1_ASeries ** s2_ASeries := Module[
   {d, k, m1, m2},
   m1 = LengthWhile[s1, # == 0 &];
   m2 = LengthWhile[s2, # == 0 &];
   ASeries @@ Table[
     Sum[s1[[k + 1]] ** s2[[d - k + 1]], {k, m1, d - m2}],
     {d, 0, Min[m1 + Length[s2] - 1, m2 + Length[s1] - 1]}
    ]
  ];
ASeries /: EulerE[s_ASeries] :=
  ASeries @@ Expand[Range[{0, 1 + Length[s]}] * (List @@ s)];
```

```
ASeries[AW[""], 0, 0, 0] + t1 + t1 ** t1 / 2 + t1 ** t1 ** t1 / 6
```

$$\text{ASeries}\left[\text{AW}[\,],\ \text{AW}[x]+\text{AW}[y],\ \frac{\text{AW}[xx]}{2}+\text{AW}[xy]+\frac{\text{AW}[yy]}{2},\ \frac{\text{AW}[xxx]}{6}+\frac{\text{AW}[xxy]}{2}+\frac{\text{AW}[xyy]}{2}+\frac{\text{AW}[yyy]}{6}\right]$$

```
ASeries[AW[""], AW["x"], AW["xx"] / 2, AW["xxx"] / 6] **
 ASeries[AW[""], AW["y"], AW["yy"] / 2, AW["yyy"] / 6]
```

$$\text{ASeries}\left[\text{AW}[\,],\ \text{AW}[x]+\text{AW}[y],\ \frac{\text{AW}[xx]}{2}+\text{AW}[xy]+\frac{\text{AW}[yy]}{2},\ \frac{\text{AW}[xxx]}{6}+\frac{\text{AW}[xxy]}{2}+\frac{\text{AW}[xyy]}{2}+\frac{\text{AW}[yyy]}{6}\right]$$

```
σ[y_LW, w_LW] /; Deg[y] == 1 := σ[y, w] = Which[
    y === w, AW[""],
    Deg[w] === 1, 0,
    True, Module[{w1, w2},
     {w1, w2} = LyndonFactorization[w];
     ι[w1] ** σ[y, w2] - ι[w2] ** σ[y, w1]
    ]
   ];
σ[y_, expr_] := Expand[expr /. w_LW :> σ[LW[y], w]] /. LieSeries → ASeries;
```

```
(# -> σ[1, #]) & /@ AllLyndonWords[{5}, {"1", "2"}]
```

```
{⟨1⟩ → AW[], ⟨2⟩ → 0, ⟨12⟩ → -AW[2], ⟨112⟩ → -2 AW[12] + AW[21], ⟨122⟩ → AW[22],
 ⟨1112⟩ → -3 AW[112] + 3 AW[121] - AW[211], ⟨1122⟩ → 2 AW[212] - AW[221],
 ⟨1222⟩ → -AW[222], ⟨11112⟩ → -4 AW[1112] + 6 AW[1121] - 4 AW[1211] + AW[2111],
 ⟨11122⟩ → -AW[1122] + 4 AW[1212] - AW[1221] - 2 AW[2121] + AW[2211],
 ⟨11212⟩ → -AW[1122] + 4 AW[1212] - AW[1221] - 3 AW[2112] + AW[2121],
 ⟨11222⟩ → -2 AW[1222] + 3 AW[2122] - 3 AW[2212] + AW[2221],
 ⟨12122⟩ → 2 AW[1222] - 3 AW[2122] + AW[2212], ⟨12222⟩ → AW[2222]}
```

```
xw = ⟨"x"⟩; yw = ⟨"y"⟩;
{σ[xw, BCH[5, xw, yw]], σ[yw, BCH[5, xw, yw]]} /. AW[s_] :> s
```

$$\Big\{\text{ASeries}\Big[,\ -\frac{y}{2},\ -\frac{xy}{6}+\frac{yx}{12}+\frac{yy}{12},\ \frac{yxy}{12}-\frac{yyx}{24},\ \frac{xxxy}{180}-\frac{xxyx}{120}-\frac{xxyy}{120}+\frac{xyxx}{180}+\frac{xyxy}{30}-\frac{xyyx}{120}+\frac{xyyy}{180}-\frac{yxxx}{720}-\frac{yxxy}{120}-\frac{yxyx}{120}-\frac{yxyy}{120}+\frac{yyxx}{180}-\frac{yyxy}{120}+\frac{yyyx}{180}-\frac{yyyy}{720}\Big],$$
$$\text{ASeries}\Big[,\ \frac{x}{2},\ \frac{xx}{12}+\frac{xy}{12}-\frac{yx}{6},\ \frac{xxy}{24}-\frac{xyx}{12},\ -\frac{xxxx}{720}+\frac{xxxy}{180}-\frac{xxyx}{120}+\frac{xxyy}{180}-\frac{xyxx}{120}-\frac{xyxy}{120}-\frac{xyyx}{120}-\frac{xyyy}{720}+\frac{yxxx}{180}-\frac{yxxy}{120}+\frac{yxyx}{30}+\frac{yxyy}{180}-\frac{yyxx}{120}-\frac{yyxy}{120}+\frac{yyyx}{180}\Big]\Big\}$$

```
σ[xw, BCH[8, xw, yw]] /. AW[s_] :→ s
```

$$\text{ASeries}\Big[,\ -\frac{y}{2},\ -\frac{xy}{6}+\frac{yx}{12}+\frac{yy}{12},\ \frac{yxy}{12}-\frac{yyx}{24},$$

$$\frac{xxxy}{180}-\frac{xxyx}{120}-\frac{xxyy}{120}+\frac{xyxx}{180}+\frac{xyxy}{30}-\frac{xyyx}{120}+\frac{xyyy}{180}-\frac{yxxx}{720}-\frac{yxxy}{120}-\frac{yxyx}{120}-$$
$$\frac{yxyy}{120}+\frac{yyxx}{180}-\frac{yyxy}{120}+\frac{yyyx}{180}-\frac{yyyy}{720},\ -\frac{yxxxy}{360}+\frac{yxxyx}{240}+\frac{yxxyy}{240}-\frac{yxyxx}{360}-\frac{yxyxy}{60}+$$
$$\frac{yxyyx}{240}-\frac{yxyyy}{360}+\frac{yyxxx}{1440}+\frac{yyxxy}{240}+\frac{yyxyx}{240}+\frac{yyxyy}{240}-\frac{yyyxx}{360}-\frac{yyyxy}{360}+\frac{yyyyx}{1440},$$
$$-\frac{xxxxxy}{5040}+\frac{xxxxyx}{2016}+\frac{xxxxyy}{2016}-\frac{xxxyxx}{1512}-\frac{xxxyxy}{630}-\frac{xxxyyx}{5040}-\frac{xxxyyy}{1512}+\frac{xxyxxx}{2016}+\frac{xxyxxy}{840}+$$
$$\frac{xxyxyx}{840}+\frac{xxyxyy}{840}-\frac{xxyyxx}{5040}+\frac{xxyyxy}{840}-\frac{xxyyyx}{5040}+\frac{xxyyyy}{2016}-\frac{xyxxxx}{5040}-\frac{xyxxxy}{630}+\frac{xyxxyx}{840}+$$
$$\frac{xyxxyy}{840}-\frac{xyxyxx}{630}-\frac{xyxyxy}{140}+\frac{xyxyyx}{840}-\frac{xyxyyy}{630}+\frac{xyyxxx}{2016}+\frac{xyyxxy}{840}+\frac{xyyxyx}{840}+\frac{xyyxyy}{840}-$$
$$\frac{xyyyxx}{1512}-\frac{xyyyxy}{630}+\frac{xyyyyx}{2016}-\frac{xyyyyy}{5040}+\frac{yxxxxx}{30240}+\frac{yxxxxy}{2016}-\frac{yxxxyx}{5040}-\frac{yxxxyy}{5040}-\frac{yxxyxx}{5040}+$$
$$\frac{yxxyxy}{840}-\frac{yxxyyx}{1120}-\frac{yxxyyy}{5040}+\frac{yxyxxx}{2016}+\frac{yxyxxy}{840}+\frac{yxyxyx}{840}+\frac{yxyxyy}{840}-\frac{yxyyxx}{5040}+\frac{yxyyxy}{840}-$$
$$\frac{yxyyyx}{5040}+\frac{yxyyyy}{2016}-\frac{yyxxxx}{5040}-\frac{yyxxxy}{5040}-\frac{yyxxyx}{1120}-\frac{yyxxyy}{1120}-\frac{yyxyxx}{5040}+\frac{yyxyxy}{840}-\frac{yyxyyx}{1120}-$$
$$\frac{yyxyyy}{5040}+\frac{yyyxxx}{3780}-\frac{yyyxxy}{5040}-\frac{yyyxyx}{5040}-\frac{yyyxyy}{5040}+\frac{yyyyxx}{3780}+\frac{yyyyxy}{2016}-\frac{yyyyyx}{5040}+\frac{yyyyyy}{30240},$$
$$\frac{yxxxxxy}{10080}-\frac{yxxxxyx}{4032}-\frac{yxxxxyy}{4032}+\frac{yxxxyxx}{3024}+\frac{yxxxyxy}{1260}+\frac{yxxxyyx}{10080}+\frac{yxxxyyy}{3024}-\frac{yxxyxxx}{4032}-$$
$$\frac{yxxyxxy}{1680}-\frac{yxxyxyx}{1680}-\frac{yxxyxyy}{1680}+\frac{yxxyyxx}{10080}-\frac{yxxyyxy}{1680}+\frac{yxxyyyx}{10080}-\frac{yxxyyyy}{4032}+$$
$$\frac{yxyxxxx}{10080}+\frac{yxyxxxy}{1260}-\frac{yxyxxyx}{1680}-\frac{yxyxxyy}{1680}+\frac{yxyxyxx}{1260}+\frac{yxyxyxy}{280}-\frac{yxyxyyx}{1680}+\frac{yxyxyyy}{1260}-$$
$$\frac{yxyyxxx}{4032}-\frac{yxyyxxy}{1680}-\frac{yxyyxyx}{1680}-\frac{yxyyxyy}{1680}+\frac{yxyyyxx}{3024}+\frac{yxyyyxy}{1260}-\frac{yxyyyyx}{4032}+\frac{yxyyyyy}{10080}-$$
$$\frac{yyxxxxx}{60480}-\frac{yyxxxxy}{4032}+\frac{yyxxxyx}{10080}+\frac{yyxxxyy}{10080}+\frac{yyxxyxx}{10080}-\frac{yyxxyxy}{1680}+\frac{yyxxyyx}{2240}+\frac{yyxxyyy}{10080}-$$
$$\frac{yyxyxxx}{4032}-\frac{yyxyxxy}{1680}-\frac{yyxyxyx}{1680}-\frac{yyxyxyy}{1680}+\frac{yyxyyxx}{10080}-\frac{yyxyyxy}{1680}+\frac{yyxyyyx}{10080}-\frac{yyxyyyy}{4032}+$$
$$\frac{yyyxxxx}{10080}+\frac{yyyxxxy}{3024}+\frac{yyyxxyx}{10080}+\frac{yyyxxyy}{10080}+\frac{yyyxyxx}{3024}+\frac{yyyxyxy}{1260}+\frac{yyyxyyx}{10080}+\frac{yyyxyyy}{3024}-$$
$$\frac{23\ yyyyxxx}{120960}-\frac{yyyyxxy}{4032}-\frac{yyyyxyx}{4032}-\frac{yyyyxyy}{4032}+\frac{yyyyyxx}{10080}+\frac{yyyyyxy}{10080}-\frac{yyyyyyx}{60480}\Big]$$